

DUM č. 18 v sadě

28. Inf-4 Jednoduchá hra Had ve Flashi (ActionScript)

Autor: Robert Havlásek

Datum: 08.06.2013

Ročník: 5AV

Anotace DUMu: Flash - příklad: Grafický layout, výpis počtu životů a počtu sněžených jablek do plochy. Výpis času. Zapauzování celé hry.\n

Materiály jsou určeny pro bezplatné používání pro potřeby výuky a vzdělávání na všech typech škol a školských zařízení. Jakékoliv další využití podléhá autorskému zákonu.



INVESTICE DO ROZVOJE VZDĚLÁVÁNÍ


Zaměříme se na komunikaci s uživatelem – nejprve budeme vypisovat stavové informace (počet životů, počet sněžených jablek, čas uplynulý od začátku levelu a od začátku hry).

Zavedeme „tlačítko“ a kl. zkratku pro zapauzování hry.

Upustíme přitom od myšlenky několika hadů v jedné ploše, stavové informace budeme psát jen k jednomu.

Životy a počet sněžených jablek v ploše

V dolních 50 pixelech plochy jsme si na začátku tvorby ponechali místo pro „stavové“ informace; do tohoto místa vyrobíme nejprve dva texty (nástrojem **A**), které v panelu Properties přepneme do „Dynamic text“ a svážeme je s nově zavedenými proměnnými plochy `textzivoty` a `textjablek`. Můžeme upravit vizuální vzhled (font, velikost, barvu, umístění textů – vše vyzkoušíme, jak bude v programu vypadat). Přímou do textů můžeme napsat přípravu: „Životy:“ a „Jablek:“, ale nemusíme, stejně se texty změní ihned při startu programu. *Pozn.: Studenti, pojmenujte „Životy“, `textzivoty` a `zivoty` stejně, budete-li mít slova různě skloňovaná („Životů“, „`text_s_zivoty`“, „`zivot`“, „`lives`“ atp.), budou se vám plést...* Například:



Někteří programátoři dělají „refresh“ stavových informací periodicky, například pravidelně při každém `onClipEvent(enterFrame)`, ale z hlediska přehledu studentů v kódu si myslím, že nezaškodí si zopakovat, kde proměnné `jablek` a `zivoty` nastavujeme a měníme – a přímo v těch místech měnit i `textjablek` a `textzivoty`.

Hlídat změny jablek je jednodušší, stačí kód hlavy upravit na dvou místech:

- Na začátku (nejlépe v `onClipEvent(load)`, kde píšeme `var jablek:Number = 0;`) je

nutno `textjablek` změnit – přidáme příkaz `_root.textjablek='Jablek: '+jablek;`

- Při sněžení jablka, v místě, kde je `jablek++`, opět `textjablek` přepíšeme – přidáme stejný příkaz `_root.textjablek='Jablek: '+jablek;`

Pedagogická poznámka: Studentům při takovýchto úpravách doporučuji vyhledat si všechny výskyty slova `jablek` (klávesou `CTRL+F`) a u každého zkusit samostatně rozhodnout, zda jde o čtení proměnné nebo o její změnu.

Hlídat změny životů je složitější, jsou na třech místech.

- Na začátku (nejlépe v `onClipEvent(load)`, kde píšeme `var zivoty:Number = 5;`) je nutno `textzivoty` změnit – přidáme příkaz `_root.textzivoty='Životy: '+zivoty;`
- Přijdeme-li o život, tedy buď v místě, kde smrt voláme, nebo přímo uvnitř smrti. Logičtější mi připadá uvnitř smrti, za místo, kde v ploše píšeme `koho.zivoty--;`, přidat stejný příkaz, tedy `_root.textzivoty='Životy: '+koho.zivoty;`
- Získáme-li bousový život za sněžená jablka, tedy v kódu hlavy, kde zatím píšeme `if ((jablek % pocetjablekprobonus) == 0) {zivoty++;trace("bonus");}` bychom místo `trace("bonus")` napsali rovněž `_root.textzivoty='Životy: '+zivoty;`

Počítání času

Vytvoříme ve vhodném místě plochy dynamický text „Čas:“, spojíme jej s (nově zavedenou) proměnnou `textcas`.

Zavedeme ale navíc proměnnou `caszacatek`, který budeme chápat jako offset (místo, odkud se čas počítá). Při startu programu bude `caszacatek` prakticky rovna nule, ale třeba při restartu hry nebo při počítání času odznovu v novém levelu se bude hodit.

Ve Flashi je přímo funkce jménem `getTimer()`, která vrací počet milisekund uběhlých od spuštění zkompilevaného programu.

Do kódu plochy, nejlépe hned na začátek, přidáme: `caszacatek=getTimer();` a hned pod něj `textcas = "Čas: "+Math.floor(getTimer()/1000)+" s";`

Až dosud jsme se v kódu plochy obešli bez neustále opakovaných příkazů. Kód plochy se provedl jednou, v hlavě jsme pak měli `onClipEvent(enterFrame)` (pro opakované příkazy) i `onClipEvent(load)` (pro jednorázové příkazy). Zde, v okamžiku počítání času, mi připadá programátorsky čistší opakující se čas napsat do plochy než do hlavy hada.

I zavedeme proto v ploše `onEnterFrame`, třeba hned na začátku:

```
onEnterFrame=function() {textcas="Čas: "+Math.floor(getTimer()/1000)+" s";}
```

Mírně složitější zápis, který počítá i s minutami:

```
onEnterFrame=function() {  
  if (getTimer()>=60000)  
    textcas="Čas: "+(Math.floor(getTimer()/1000/60))+" m "+  
              (Math.floor(getTimer()/1000)%60)+" s";  
  else textcas="Čas: "+Math.floor(getTimer()/1000)+" s";  
}
```

Zapauzování hry

Vytvoříme ve vhodném místě plochy nápis s vhodným textem (např. „P=Pauza“), převedeme jej do Symbolu typu `MovieClip` s instanceName „pauza“ a spojíme jeho událost `on (press)` s zapauzováním a odpauzováním celé hry. Podobně bychom ošetřili i stisk klávesy P, to je ale lépe udělat pomocí objektu `Key` v `onClipEvent(enterFrame)`. Mají-li všechny události vyústit ve stejný kód, vyrobil bych funkci (nazvěme ji třeba `pauza`), kterou zavolám z každé obsluhy:

```
on (press) {pauza();}  
onClipEvent(enterFrame) {if (Key.isDown(80)) pauza();}  
// kod 80 odpovida fyzicke klavese P  
onClipEvent(load) {  
  pauza=function() { smer=5; }  
}
```

Vidíte, že zapauzovat hada znamená nastavit mu jeho směr třeba na 5 (hodnoty 0, 1, 2, 3 jsou běžné směry, 4 je stojící nezapauzovaný had).

Hru spustíme, vyzkoušíme stisknout P nebo kliknout na nápis. Vidíme zásadní komplikaci: hlava se sice zastaví, ale tělo doběhne pod ni. To lze jednoduše opravit – v kódu hlavy necháme tělo rovněž refreshovat jen v případě, že bude směr <4:

```
// VYPOCET NOVEHO TELA:
if (_root.hlava.smer<4)
    for (m=telox.length-2;m>=0;m--) {telox[m+1]=telox[m];
                                     teloy[m+1]=teloy[m];}
```

Hra je sice zapauzována, had se nehýbe, ale čas se nám počítá dál... Jako zadání do budoucna můžeme promyslet, jak by bylo komplikované zastavit při pauze počítání času. (Museli bychom si pamatovat začátek i konec pauzy a jejich rozdíl přidat k proměnné caszacatek).

Rovněž jako dobrovolné bych ponechal grafické ztvárnění pauzy, například poloprůhledným šedým obdélníkem, kterému nejprve nastavíme vlastnost visible na false, při začátku pauzy naopak na true a odpauzování jej zase zneviditelníme.

Výsledné kódy plochy, hlavy a textu s pauzou

Nejprve kód plochy:

```
onEnterFrame=function() {
if (getTimer())>=60000)
    textcas="Čas: "+(Math.floor(getTimer()/1000/60))+" m "+
              (Math.floor(getTimer()/1000)%60)+" s";
    else textcas="Čas: "+Math.floor(getTimer()/1000)+" s";
}

caszacatek=getTimer();
textcas = "Čas: "+Math.floor(getTimer()/1000)+" s";
var zdix:Array = Array();
var zdiy:Array = Array();
var klonzdi:Array = Array();
// TVORBA POLE ZDI OKOLO:
for (i=0; i<=310; i=i+10) {zdix.push(i);zdiy.push(0);} // horni rada
for (i=0; i<=310; i=i+10) {zdix.push(i);zdiy.push(190);} // spodni rada
for (i=0; i<=190; i=i+10) {zdix.push(0);zdiy.push(i);} // leva rada
for (i=0; i<=190; i=i+10) {zdix.push(310);zdiy.push(i);} // prava rada
// FYZICKE KLONOVANI ZDI OKOLO:
for (k=0; k<zdix.length; k++){
    _root.vzorzdi._x=zdix[k];
    _root.vzorzdi._y=zdiy[k];
    duplicateMovieClip(_root.vzorzdi,"klonzdi["+k+"]",50+getNextHighestDepth());
}
// NAHODNE ZDI A JEJICH KONTROLA:
for (k=0; k<20; k=k+1){
    _root.vzorzdi._x=10+10*random(30);
    _root.vzorzdi._y=10+10*random(18);
    // KONTROLA OBSAZENOSTI:
    var obsazeno=0;
    for (m=0; m<zdix.length; m=m+1)
        if ((_root.vzorzdi._x==zdix[m])&&(_root.vzorzdi._y==zdiy[m]))
            {obsazeno=1;break;}
    if ( (obsazeno==1) or
        ((_root.vzorzdi._x==160)&&(_root.vzorzdi._y==100)) ) {k=k-1}
    else {zdix.push(_root.vzorzdi._x);
          zdiy.push(_root.vzorzdi._y);
          duplicateMovieClip(_root.vzorzdi,"klonzdi["+zdix.length-1+"]",ge
tNextHighestDepth());}
}
```

```

tvor_jablko("jablko1",41);

function tvor_jablko(jmeno:String, hloubka:Number):Number {
    // TVORBA JABLKA:
    this.createEmptyMovieClip(jmeno, hloubka);
    var j=this.getInstanceAtDepth(hloubka);
        j.beginFill(0xFF0000);

    drawCircle(j, 5, 5, 5);
    j.endFill();
    do {
        j._x=10+10*random(30);
        j._y=10+10*random(18);
        var obsazeno=0;
        for (m=0; m<zdix.length; m=m+1)
            if ((j._x==zdix[m]) && (j._y==zdiy[m]))
                {obsazeno=1;break;}
        for (m=0; m<_root.hlava.telox.length; m=m+1)
            if ((j._x==_root.hlava.telox[m]) && (j._y==_root.hlava.teloy[m]))
                {obsazeno=1;break;}
        } while (obsazeno==1);
    return 0;
} // konec funkce tvor_jablko

function drawCircle(mc:MovieClip, x:Number, y:Number, r:Number):Void {
    mc.moveTo(x+r, y);
    mc.curveTo(r+x, Math.tan(Math.PI/8)*r+y, Math.sin(Math.PI/4)*r+x,
Math.sin(Math.PI/4)*r+y);
    mc.curveTo(Math.tan(Math.PI/8)*r+x, r+y, x, r+y);
    mc.curveTo(-Math.tan(Math.PI/8)*r+x, r+y, -Math.sin(Math.PI/4)*r+x,
Math.sin(Math.PI/4)*r+y);
    mc.curveTo(-r+x, Math.tan(Math.PI/8)*r+y, -r+x, y);
    mc.curveTo(-r+x, -Math.tan(Math.PI/8)*r+y, -Math.sin(Math.PI/4)*r+x,
-Math.sin(Math.PI/4)*r+y);
    mc.curveTo(-Math.tan(Math.PI/8)*r+x, -r+y, x, -r+y);
    mc.curveTo(Math.tan(Math.PI/8)*r+x, -r+y, Math.sin(Math.PI/4)*r+x,
-Math.sin(Math.PI/4)*r+y);
    mc.curveTo(r+x, -Math.tan(Math.PI/8)*r+y, r+x, y);
} // konec funkce drawCircle

function smrt(koho:Object) {
    koho.zivoty--;_root.textzivoty='Životy: '+koho.zivoty;
    koho._x=160;
    koho._y=100;
    koho.smer=4;
    for (var m=0;m<koho.telox.length;m++)
        {koho.telox[m]=koho._x;
        koho.teloy[m]=koho._y;
        }
}

```

v ploše:

Následuje kód textu s pauzou:

```

on (press) {pauza();}
onClipEvent(enterFrame) {if (Key.isDown(80)) pauza();}
// kod 80 odpovida fyzicke klavesce P
onClipEvent(load) {
    pauza=function() { smer=5; }
}

```

v textu s pauzou

A též kód hlavy:

```
onClipEvent(enterFrame)
{
    if (Key.isDown(Key.LEFT)) smer=2;
    if (Key.isDown(Key.RIGHT)) smer=0;
    if (Key.isDown(Key.UP)) smer=1;
    if (Key.isDown(Key.DOWN)) smer=3;
    if (smer==0) {_root.hlava._x = _root.hlava._x + 10;}
    if (smer==1) {_root.hlava._y = _root.hlava._y - 10;}
    if (smer==2) {_root.hlava._x = _root.hlava._x - 10;}
    if (smer==3) {_root.hlava._y = _root.hlava._y + 10;}
    if (_root.hlava._x>=320) {_root.hlava._x=0;}
    if (_root.hlava._y>=200) {_root.hlava._y=0;}
    if (_root.hlava._x<0) {_root.hlava._x=310;}
    if (_root.hlava._y<0) {_root.hlava._y=190;}
    // KONTROLA, ZDA PRAVE NEJIME JABLKO:
    if ((_root.hlava._x==_root.jablkol._x)&&(_root.hlava._y==_root.jabl
kol._y))
        {jablek++;_root.textjablek='Jablek: '+jablek;
        if ((jablek % pocetjablekprobonus) == 0)
            {zivoty++;_root.textzivoty='Životy: '+zivoty;}
        _root.tvor_jablko("jablko1",41);
        var p=_root.hlava.telox.length;
        _root.hlava.telox.push(_root.hlava.telox[p]);
        _root.hlava.teloy.push(_root.hlava.teloy[p]);
        }
    // KONTROLA NARAZENI HLAVY DO ZDI:
    var obsazeno=0;
    for (m=0; m<_root.zdix.length; m=m+1)
        if ((_root.hlava._x==_root.zdix[m])&&(_root.hlava._y==_root.zdiy[m]))
            {obsazeno=1;break;}
    // A I KONTROLA NARAZENI DO SVEHO TELA (JEN POKUD SE HYBEME):
    if (smer<4) for (m=0; m<_root.hlava.telox.length; m=m+1)
        if ((_root.hlava._x==_root.hlava.telox[m])&&(_root.hlava._y==_root.hla
va.teloy[m]))
            {obsazeno=1;break;}
    if (obsazeno==1) _root.smrt(this);
/*    if ((_root.hlava._x>=320) || (_root.hlava._y>=200) ||
        (_root.hlava._x<0) || (_root.hlava._y<0)) _root.smrt();*/
    // VYPOCET NOVEHO TELA:
    if (_root.hlava.smer<4)
        for (m=telox.length-2;m>=0;m--) {telox[m+1]=telox[m];
            teloy[m+1]=teloy[m];}

    telox[0]=_root.hlava._x;
    teloy[0]=_root.hlava._y;
    // KRESLENI TELA:
    for (var m=0;m<telox.length;m++)
        {removeMovieClip("klontelo["+m+"]");
        _root.telovzor._x=telox[m];
        _root.telovzor._y=teloy[m];
        duplicateMovieClip(_root.telovzor,"klontelo["+m+"]",m)
        }
} // konec onClipEvent(enterFrame)

onClipEvent(load) {
    var telox:Array = Array();
    var teloy:Array = Array();
    var defaultnidelka:Number = 4;
    var zivoty:Number = 5;_root.textzivoty='Životy: '+zivoty;
    var jablek:Number = 0;_root.textjablek='Jablek: '+jablek;
    var pocetjablekprobonus:Number = 5;
```

v objektu hlava:

```
_root.telovzor._x=_root.hlava._x; //posuneme vzor do pozice hlavy
_root.telovzor._y=_root.hlava._y;
for (var i=0; i<defaultnidelka; i++) //zduplikujeme (defaultnidelka)-krat
{telox.push(_root.hlava._x);
  teloy.push(_root.hlava._y);
  duplicateMovieClip(_root.telovzor,"klontelo"+i+"",i)
} //konec cyklu for
} // konec onClipEvent(load)
```